Föreläsning 2

Differentialekvationer och Maple

2.1 Aktuella avsnitt i läroboken

(1.3) Slope Fields and Solution Curves.

Se också maplehäftet [7].

För den som är intresserad av numeriska lösningsmetoder för differentialekvationer kan följande avsnitt vara intressanta. De ingår dock inte i kursen.

(2.4) Numerical Approximation: Euler's Method.

(2.5) A Closer Look at the Euler Method.

(2.6) The Runge-Kutta Method.

Följande exempel bör man sitta framför datorn och köra i Maple.

2.2 Funktionen `int`

Den enklaste typen av differentialekvationer av formen $y' = f(x)$ med lösningen $y(x) = \int f(x)\,dx + C$ kan man hantera med funktionen `int` som ger primitiva funktioner.

2.2.1 Kanalsimningen

För detaljerna se exempel 4. Högerledet är en funktion f

```
> f := x -> (v[0]/v[s])*(1-x^2/a^2);
```

$$f := x \to \frac{v_0\,(1 - \frac{x^2}{a^2})}{v_s}$$

En primitiv funktion är

```
> sol_1 := int(f(x), x);
```

$$sol\_1 := \frac{v_0\,(x - \frac{x^3}{3\,a^2})}{v_s}$$

men eftersom vi vill ha en lösning som uppfyller begynnelsevillkoret $y(-a) = 0$, är det bättre att använda den bestämda integralen $y(x) = \int_{-a}^{x} f(t)\,dt$

```
> sol_2 := int(f(t), t=-a..x);
```

$$sol\_2 := -\frac{1}{3}\frac{v_0\,(x^3 + a^3)}{v_s\,a^2} + \frac{v_0\,(x + a)}{v_s}$$

Lösningen kan förenklas något

```
> sol_2 := simplify(sol_2);
```

$$sol\_2 := -\frac{1}{3}\frac{v_0\,(x^3 - 2\,a^3 - 3\,a^2\,x)}{v_s\,a^2}$$

Lösningens värde för $x = a$ kan beräknas med funktionen `eval`

```
> eval(sol_2, x=a);
```

$$\frac{4}{3}\frac{v_0\,a}{v_s}$$

Det är emellertid ofta bättre att definiera lösningen som en funktion $y(x)$. Detta åstadkommer funktionen `unapply`

```
> y := unapply(sol_2,x);
```

$$y := x \to -\frac{1}{3}\,\frac{v_0\,(x^3 - 2\,a^3 - 3\,a^2\,x)}{v_s\,a^2}$$

Slutligen ger vi de ingående konstanterna numeriska värden och använder `plot`-funktionen för att rita lösningskurvan $y(x)$ för $-1/2 \le x \le 1/2$. Se figur 2.1.

```
> a := 1/2; v[0] := 9; v[s] := 3;
```

$$a := \frac{1}{2}$$
$$v_0 := 9$$
$$v_s := 3$$



Figur 2.1: Funktionen $y = 1 + 3x - 4x^3$.

2.3 Funktionen `dsolve`

Många differentialekvationer av standardtyp, speciellt alla som vi behandlar i den här kursen, känner Maple till och kan lösa symboliskt. Den funktion som används heter `dsolve`.

2.3.1 Torricellis tratt

Se exempel 6. Differentialekvationen matas in på följande sätt. Observera att y-variabeln skrivs som $y(t)$.

```
> ode := diff(y(t),t) =
> -r^2*sqrt(2*g*y(t))/(r+(R-r)*y(t)/h)^2;
```

$$ode := \frac{d}{dt}\,\mathrm{y}(t) = -\frac{r^2\,\sqrt{2}\,\sqrt{g\,\mathrm{y}(t)}}{(r + \frac{(R-r)\,\mathrm{y}(t)}{h})^2}$$

Funktionen y har ett ganska komplicerat uttryck och Maple väljer att ge lösningen i implicit form.

```
> sol_1 := dsolve(ode, y(t));
```

$$sol\_1 := t + \frac{\sqrt{2}\left(\frac{1}{5}(-R+r)^2\,(g\,y(t))^{(5/2)} - \frac{2}{3}\,r\,h\,g\,(-R+r)\,(g\,y(t))^{(3/2)} + r^2\,h^2\,g^2\,\sqrt{g\,y(t)}\right)}{r^2\,h^2\,g^3} + \_C1 = 0$$

Integrationskonstanten bestäms av att $y(t) = h$ då $t = 0$. Vi får ekvationen

```
> eq_1 := eval(sol_1, {t = 0, y(t) = h});
```

$$eq\_1 := \frac{\sqrt{2}\left(\frac{(-R+r)^2\,(g\,h)^{(5/2)}}{5} - \frac{2\,r\,h\,g\,(-R+r)\,(g\,h)^{(3/2)}}{3} + r^2\,h^2\,g^2\,\sqrt{g\,h}\right)}{r^2\,h^2\,g^3} + \_C1 = 0$$

som har lösningen

```
> ic := solve(eq_1, _C1);
```

$$ic := -\frac{\sqrt{2}\,\sqrt{g\,h}\,(3\,R^2 + 4\,R\,r + 8\,r^2)}{15\,g\,r^2}$$

Substituerar vi det erhållna värdet på integrationskonstanten får vi

```
> sol_2 := eval(sol_1, _C1 = ic);
```

$$sol\_2 := t + \frac{\sqrt{2}\left(\frac{1}{5}(-R+r)^2\,(g\,y(t))^{(5/2)} - \frac{2}{3}\,r\,h\,g\,(-R+r)\,(g\,y(t))^{(3/2)} + r^2\,h^2\,g^2\,\sqrt{g\,y(t)}\right)}{r^2\,h^2\,g^3} - \frac{\sqrt{2}\,\sqrt{g\,h}\,(3\,R^2 + 4\,R\,r + 8\,r^2)}{15\,g\,r^2} = 0$$

Tömningstiden T ges av villkoret: $y(t) = 0$ för $t = T$ vilket ger ekvationen

```
> eq_2 := eval(sol_2, {y(t) = 0, t = T});
```

$$eq\_2 := T - \frac{\sqrt{2}\,\sqrt{g\,h}\,(3\,R^2 + 4\,R\,r + 8\,r^2)}{15\,g\,r^2} = 0$$

med lösningen

```
> T := solve(eq_2, T);
```

$$T := \frac{\sqrt{2}\,\sqrt{g\,h}\,(3\,R^2 + 4\,R\,r + 8\,r^2)}{15\,g\,r^2}$$

Numeriska värden:

```
> g := 9.82; h := 0.12; R := 15*r;
```

$$g := 9.82$$
$$h := 0.12$$
$$R := 15\,r$$

Tiden T med 3 siffror blir då

```
> evalf(T, 3);
```

$$7.76$$

2.4 Riktningsfält och lösningskurvor

Om ekvationen $Mdx+Ndy=0$ är exakt ges lösningskurvorna av nivåkurvor $F(x,y)=C$ till potentialen F, där $\nabla F=(M,N)$. Nivåkurvor ritars i Maple med funktionen `contourplot`.

2.4.1 Nivåkurvor

Ekvationen i exempel 13 hanteras i Maple på följande sätt: M och N matas in och vi kontrollerar att ekvationen är exakt

```
> M := 8*x-10+y; N := 32/9*y-73/9+x;
```

$$M := 8x-10+y$$

$$N := \frac{32y}{9}-\frac{73}{9}+x$$

```
> diff(M,y)-diff(N,x);
```

$$0$$

Potentialfunktionen F fås genom integration med avseende på y. Observera att ”integrationskonstanten” $f(x)$ är en funktion av x.

```
> F := int(N,y)+f(x);
```

$$F := \frac{16y^2}{9}-\frac{73y}{9}+xy+\mathrm{f}(x)$$

Funktionen f satisfierar differentialekvationen

```
> ode := diff(F,x) - M = 0;
```

$$ode := \left(\frac{d}{dx}\mathrm{f}(x)\right)-8x+10=0$$

med lösningen

```
> sol := dsolve(ode, f(x));
```

$$sol := \mathrm{f}(x)=4x^2-10x+\_C1$$

Vi uppdaterar F med den funna lösningen

```
> F := eval(F, sol);
```

$$F := \frac{16}{9}y^2-\frac{73}{9}y+xy+4x^2-10x+\_C1$$

och sätter sedan konstanten till noll

```
> F := eval(F, _C1 = 0);
```

$$F := \frac{16}{9}y^2 - \frac{73}{9}y + xy + 4x^2 - 10x$$

Funktionen contourplot finns i paketet plots som först måste laddas in

```
> with(plots):

Warning, the name changecoords has been redefined
```

Man får sedan prova sig fram till lämpliga värden på de olika parametrarna.

```
> contourplot(F, x=-10..10, y=-10..10, grid=[200,200],
> contours=10, thickness=3);
```



Figur 2.2: Nivåkurvor till F.

2.4.2 Riktningsfält

Lösningskurvorna till $Mdx + Ndy = 0$ är ortogonala mot vektorfältet (M,N) och därmed tangentiella mot $(-N,M)$ som kallas ekvationens *riktningsfält*. Plana vektorfält kan Matlab rita med funktionen `fieldplot` som också finns i plot-paketet. Med exemplet 6 får vi: